// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby disabling any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/token/ERC20/ERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (token/ERC20/ERC20.sol)

pragma solidity ^0.8.0;

import "./IERC20.sol";
import "./extensions/IERC20Metadata.sol";
import "../../utils/Context.sol";

/**
 * @dev Implementation of the {IERC20} interface.
 *
 * This implementation is agnostic to the way tokens are created. This means
 * that a supply mechanism has to be added in a derived contract using {_mint}.
 * For a generic mechanism see {ERC20PresetMinterPauser}.
 *
 * TIP: For a detailed writeup see our guide
 * https://forum.openzeppelin.com/t/how-to-implement-erc20-supply-mechanisms/226[How
 * to implement supply mechanisms].
 *
 * The default value of {decimals} is 18. To change this, you should override
 * this function so it returns a different value.
 *
 * We have followed general OpenZeppelin Contracts guidelines: functions revert
 * instead returning `false` on failure. This behavior is nonetheless
 * conventional and does not conflict with the expectations of ERC20
 * applications.
 *
 * Additionally, an {Approval} event is emitted on calls to {transferFrom}.
 * This allows applications to reconstruct the allowance for all accounts just
 * by listening to said events. Other implementations of the EIP may not emit
 * these events, as it isn't required by the specification.
 *
 * Finally, the non-standard {decreaseAllowance} and {increaseAllowance}
 * functions have been added to mitigate the well-known issues around setting
 * allowances. See {IERC20-approve}.
 */
contract ERC20 is Context, IERC20, IERC20Metadata {
    mapping(address => uint256) private _balances;

    mapping(address => mapping(address => uint256)) private _allowances;

    uint256 private _totalSupply;

    string private _name;
    string private _symbol;

    /**
     * @dev Sets the values for {name} and {symbol}.
     *
     * All two of these values are immutable: they can only be set once during
     * construction.
     */
    constructor(string memory name_, string memory symbol_) {
        _name = name_;
        _symbol = symbol_;
    }

    /**
     * @dev Returns the name of the token.
     */
    function name() public view virtual override returns (string memory) {
        return _name;
    }

    /**
     * @dev Returns the symbol of the token, usually a shorter version of the
     * name.
     */
    function symbol() public view virtual override returns (string memory) {
        return _symbol;
    }

    /**
     * @dev Returns the number of decimals used to get its user representation.
     * For example, if `decimals` equals `2`, a balance of `505` tokens should
     * be displayed to a user as `5.05` (`505 / 10 ** 2`).
     *
     * Tokens usually opt for a value of 18, imitating the relationship between
     * Ether and Wei. This is the default value returned by this function, unless
     * it's overridden.
     *
     * NOTE: This information is only used for _display_ purposes: it in
     * no way affects any of the arithmetic of the contract, including
     * {IERC20-balanceOf} and {IERC20-transfer}.
     */
    function decimals() public view virtual override returns (uint8) {
        return 18;
    }

    /**
     * @dev See {IERC20-totalSupply}.
     */
    function totalSupply() public view virtual override returns (uint256) {
        return _totalSupply;
    }

    /**
     * @dev See {IERC20-balanceOf}.
     */
    function balanceOf(address account) public view virtual override returns (uint256) {
        return _balances[account];
    }

    /**
     * @dev See {IERC20-transfer}.
     *
     * Requirements:
     *
     * - `to` cannot be the zero address.
     * - the caller must have a balance of at least `amount`.
     */
    function transfer(address to, uint256 amount) public virtual override returns (bool) {
        address owner = _msgSender();
        _transfer(owner, to, amount);
        return true;
    }

    /**
     * @dev See {IERC20-allowance}.
     */
    function allowance(address owner, address spender) public view virtual override returns (uint256) {
        return _allowances[owner][spender];
    }

    /**
     * @dev See {IERC20-approve}.
     *
     * NOTE: If `amount` is the maximum `uint256`, the allowance is not updated on
     * `transferFrom`. This is semantically equivalent to an infinite approval.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     */
    function approve(address spender, uint256 amount) public virtual override returns (bool) {
        address owner = _msgSender();
        _approve(owner, spender, amount);
        return true;
    }

    /**
     * @dev See {IERC20-transferFrom}.
     *
     * Emits an {Approval} event indicating the updated allowance. This is not
     * required by the EIP. See the note at the beginning of {ERC20}.
     *
     * NOTE: Does not update the allowance if the current allowance
     * is the maximum `uint256`.
     *
     * Requirements:
     *
     * - `from` and `to` cannot be the zero address.
     * - `from` must have a balance of at least `amount`.
     * - the caller must have allowance for ``from``'s tokens of at least
     * `amount`.
     */
    function transferFrom(address from, address to, uint256 amount) public virtual override returns (bool) {
        address spender = _msgSender();
        _spendAllowance(from, spender, amount);
        _transfer(from, to, amount);
        return true;
    }

    /**
     * @dev Atomically increases the allowance granted to `spender` by the caller.
     *
     * This is an alternative to {approve} that can be used as a mitigation for
     * problems described in {IERC20-approve}.
     *
     * Emits an {Approval} event indicating the updated allowance.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     */
    function increaseAllowance(address spender, uint256 addedValue) public virtual returns (bool) {
        address owner = _msgSender();
        _approve(owner, spender, allowance(owner, spender) + addedValue);
        return true;
    }

    /**
     * @dev Atomically decreases the allowance granted to `spender` by the caller.
     *
     * This is an alternative to {approve} that can be used as a mitigation for
     * problems described in {IERC20-approve}.
     *
     * Emits an {Approval} event indicating the updated allowance.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     * - `spender` must have allowance for the caller of at least
     * `subtractedValue`.
     */
    function decreaseAllowance(address spender, uint256 subtractedValue) public virtual returns (bool) {
        address owner = _msgSender();
        uint256 currentAllowance = allowance(owner, spender);
        require(currentAllowance >= subtractedValue, "ERC20: decreased allowance below zero");
        unchecked {
            _approve(owner, spender, currentAllowance - subtractedValue);
        }

        return true;
    }

    /**
     * @dev Moves `amount` of tokens from `from` to `to`.
     *
     * This internal function is equivalent to {transfer}, and can be used to
     * e.g. implement automatic token fees, slashing mechanisms, etc.
     *
     * Emits a {Transfer} event.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `from` must have a balance of at least `amount`.
     */
    function _transfer(address from, address to, uint256 amount) internal virtual {
        require(from != address(0), "ERC20: transfer from the zero address");
        require(to != address(0), "ERC20: transfer to the zero address");

        _beforeTokenTransfer(from, to, amount);

        uint256 fromBalance = _balances[from];
        require(fromBalance >= amount, "ERC20: transfer amount exceeds balance");
        unchecked {
            _balances[from] = fromBalance - amount;
            // Overflow not possible: the sum of all balances is capped by totalSupply, and the sum is preserved by
            // decrementing then incrementing.
            _balances[to] += amount;
        }

        emit Transfer(from, to, amount);

        _afterTokenTransfer(from, to, amount);
    }

    /** @dev Creates `amount` tokens and assigns them to `account`, increasing
     * the total supply.
     *
     * Emits a {Transfer} event with `from` set to the zero address.
     *
     * Requirements:
     *
     * - `account` cannot be the zero address.
     */
    function _mint(address account, uint256 amount) internal virtual {
        require(account != address(0), "ERC20: mint to the zero address");

        _beforeTokenTransfer(address(0), account, amount);

        _totalSupply += amount;
        unchecked {
            // Overflow not possible: balance + amount is at most totalSupply + amount, which is checked above.
            _balances[account] += amount;
        }
        emit Transfer(address(0), account, amount);

        _afterTokenTransfer(address(0), account, amount);
    }

    /**
     * @dev Destroys `amount` tokens from `account`, reducing the
     * total supply.
     *
     * Emits a {Transfer} event with `to` set to the zero address.
     *
     * Requirements:
     *
     * - `account` cannot be the zero address.
     * - `account` must have at least `amount` tokens.
     */
    function _burn(address account, uint256 amount) internal virtual {
        require(account != address(0), "ERC20: burn from the zero address");

        _beforeTokenTransfer(account, address(0), amount);

        uint256 accountBalance = _balances[account];
        require(accountBalance >= amount, "ERC20: burn amount exceeds balance");
        unchecked {
            _balances[account] = accountBalance - amount;
            // Overflow not possible: amount <= accountBalance <= totalSupply.
            _totalSupply -= amount;
        }

        emit Transfer(account, address(0), amount);

        _afterTokenTransfer(account, address(0), amount);
    }

    /**
     * @dev Sets `amount` as the allowance of `spender` over the `owner` s tokens.
     *
     * This internal function is equivalent to `approve`, and can be used to
     * e.g. set automatic allowances for certain subsystems, etc.
     *
     * Emits an {Approval} event.
     *
     * Requirements:
     *
     * - `owner` cannot be the zero address.
     * - `spender` cannot be the zero address.
     */
    function _approve(address owner, address spender, uint256 amount) internal virtual {
        require(owner != address(0), "ERC20: approve from the zero address");
        require(spender != address(0), "ERC20: approve to the zero address");

        _allowances[owner][spender] = amount;
        emit Approval(owner, spender, amount);
    }

    /**
     * @dev Updates `owner` s allowance for `spender` based on spent `amount`.
     *
     * Does not update the allowance amount in case of infinite allowance.
     * Revert if not enough allowance is available.
     *
     * Might emit an {Approval} event.
     */
    function _spendAllowance(address owner, address spender, uint256 amount) internal virtual {
        uint256 currentAllowance = allowance(owner, spender);
        if (currentAllowance != type(uint256).max) {
            require(currentAllowance >= amount, "ERC20: insufficient allowance");
            unchecked {
                _approve(owner, spender, currentAllowance - amount);
            }
        }
    }

    /**
     * @dev Hook that is called before any transfer of tokens. This includes
     * minting and burning.
     *
     * Calling conditions:
     *
     * - when `from` and `to` are both non-zero, `amount` of ``from``'s tokens
     * will be transferred to `to`.
     * - when `from` is zero, `amount` tokens will be minted for `to`.
     * - when `to` is zero, `amount` of ``from``'s tokens will be burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _beforeTokenTransfer(address from, address to, uint256 amount) internal virtual {}

    /**
     * @dev Hook that is called after any transfer of tokens. This includes
     * minting and burning.
     *
     * Calling conditions:
     *
     * - when `from` and `to` are both non-zero, `amount` of ``from``'s tokens
     * has been transferred to `to`.
     * - when `from` is zero, `amount` tokens have been minted for `to`.
     * - when `to` is zero, `amount` of ``from``'s tokens have been burned.
     * - `from` and `to` are never both zero.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _afterTokenTransfer(address from, address to, uint256 amount) internal virtual {}
}

// File: @openzeppelin/contracts/token/ERC20/extensions/ERC20Burnable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.5.0) (token/ERC20/extensions/ERC20Burnable.sol)

pragma solidity ^0.8.0;

import "../ERC20.sol";
import "../../../utils/Context.sol";

/**
 * @dev Extension of {ERC20} that allows token holders to destroy both their own
 * tokens and those that they have an allowance for, in a way that can be
 * recognized off-chain (via event analysis).
 */
abstract contract ERC20Burnable is Context, ERC20 {
    /**
     * @dev Destroys `amount` tokens from the caller.
     *
     * See {ERC20-_burn}.
     */
    function burn(uint256 amount) public virtual {
        _burn(_msgSender(), amount);
    }

    /**
     * @dev Destroys `amount` tokens from `account`, deducting from the caller's
     * allowance.
     *
     * See {ERC20-_burn} and {ERC20-allowance}.
     *
     * Requirements:
     *
     * - the caller must have allowance for ``accounts``'s tokens of at least
     * `amount`.
     */
    function burnFrom(address account, uint256 amount) public virtual {
        _spendAllowance(account, _msgSender(), amount);
        _burn(account, amount);
    }
}

// File: @openzeppelin/contracts/token/ERC20/extensions/IERC20Metadata.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC20/extensions/IERC20Metadata.sol)

pragma solidity ^0.8.0;

import "../IERC20.sol";

/**
 * @dev Interface for the optional metadata functions from the ERC20 standard.
 *
 * _Available since v4.1._
 */
interface IERC20Metadata is IERC20 {
    /**
     * @dev Returns the name of the token.
     */
    function name() external view returns (string memory);

    /**
     * @dev Returns the symbol of the token.
     */
    function symbol() external view returns (string memory);

    /**
     * @dev Returns the decimals places of the token.
     */
    function decimals() external view returns (uint8);
}

// File: @openzeppelin/contracts/token/ERC20/extensions/IERC20Permit.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (token/ERC20/extensions/IERC20Permit.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 Permit extension allowing approvals to be made via signatures, as defined in
 * https://eips.ethereum.org/EIPS/eip-2612[EIP-2612].
 *
 * Adds the {permit} method, which can be used to change an account's ERC20 allowance (see {IERC20-allowance}) by
 * presenting a message signed by the account. By not relying on {IERC20-approve}, the token holder account doesn't
 * need to send a transaction, and thus is not required to hold Ether at all.
 */
interface IERC20Permit {
    /**
     * @dev Sets `value` as the allowance of `spender` over ``owner``'s tokens,
     * given ``owner``'s signed approval.
     *
     * IMPORTANT: The same issues {IERC20-approve} has related to transaction
     * ordering also apply here.
     *
     * Emits an {Approval} event.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     * - `deadline` must be a timestamp in the future.
     * - `v`, `r` and `s` must be a valid `secp256k1` signature from `owner`
     * over the EIP712-formatted function arguments.
     * - the signature must use ``owner``'s current nonce (see {nonces}).
     *
     * For more information on the signature format, see the
     * https://eips.ethereum.org/EIPS/eip-2612#specification[relevant EIP
     * section].
     */
    function permit(
        address owner,
        address spender,
        uint256 value,
        uint256 deadline,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) external;

    /**
     * @dev Returns the current nonce for `owner`. This value must be
     * included whenever a signature is generated for {permit}.
     *
     * Every successful call to {permit} increases ``owner``'s nonce by one. This
     * prevents a signature from being used multiple times.
     */
    function nonces(address owner) external view returns (uint256);

    /**
     * @dev Returns the domain separator used in the encoding of the signature for {permit}, as defined by {EIP712}.
     */
    // solhint-disable-next-line func-name-mixedcase
    function DOMAIN_SEPARATOR() external view returns (bytes32);
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address from, address to, uint256 amount) external returns (bool);
}

// File: @openzeppelin/contracts/token/ERC20/utils/SafeERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (token/ERC20/utils/SafeERC20.sol)

pragma solidity ^0.8.0;

import "../IERC20.sol";
import "../extensions/IERC20Permit.sol";
import "../../../utils/Address.sol";

/**
 * @title SafeERC20
 * @dev Wrappers around ERC20 operations that throw on failure (when the token
 * contract returns false). Tokens that return no value (and instead revert or
 * throw on failure) are also supported, non-reverting calls are assumed to be
 * successful.
 * To use this library you can add a `using SafeERC20 for IERC20;` statement to your contract,
 * which allows you to call the safe operations as `token.safeTransfer(...)`, etc.
 */
library SafeERC20 {
    using Address for address;

    /**
     * @dev Transfer `value` amount of `token` from the calling contract to `to`. If `token` returns no value,
     * non-reverting calls are assumed to be successful.
     */
    function safeTransfer(IERC20 token, address to, uint256 value) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transfer.selector, to, value));
    }

    /**
     * @dev Transfer `value` amount of `token` from `from` to `to`, spending the approval given by `from` to the
     * calling contract. If `token` returns no value, non-reverting calls are assumed to be successful.
     */
    function safeTransferFrom(IERC20 token, address from, address to, uint256 value) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transferFrom.selector, from, to, value));
    }

    /**
     * @dev Deprecated. This function has issues similar to the ones found in
     * {IERC20-approve}, and its usage is discouraged.
     *
     * Whenever possible, use {safeIncreaseAllowance} and
     * {safeDecreaseAllowance} instead.
     */
    function safeApprove(IERC20 token, address spender, uint256 value) internal {
        // safeApprove should only be called when setting an initial allowance,
        // or when resetting it to zero. To increase and decrease it, use
        // 'safeIncreaseAllowance' and 'safeDecreaseAllowance'
        require(
            (value == 0) || (token.allowance(address(this), spender) == 0),
            "SafeERC20: approve from non-zero to non-zero allowance"
        );
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, value));
    }

    /**
     * @dev Increase the calling contract's allowance toward `spender` by `value`. If `token` returns no value,
     * non-reverting calls are assumed to be successful.
     */
    function safeIncreaseAllowance(IERC20 token, address spender, uint256 value) internal {
        uint256 oldAllowance = token.allowance(address(this), spender);
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, oldAllowance + value));
    }

    /**
     * @dev Decrease the calling contract's allowance toward `spender` by `value`. If `token` returns no value,
     * non-reverting calls are assumed to be successful.
     */
    function safeDecreaseAllowance(IERC20 token, address spender, uint256 value) internal {
        unchecked {
            uint256 oldAllowance = token.allowance(address(this), spender);
            require(oldAllowance >= value, "SafeERC20: decreased allowance below zero");
            _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, oldAllowance - value));
        }
    }

    /**
     * @dev Set the calling contract's allowance toward `spender` to `value`. If `token` returns no value,
     * non-reverting calls are assumed to be successful. Compatible with tokens that require the approval to be set to
     * 0 before setting it to a non-zero value.
     */
    function forceApprove(IERC20 token, address spender, uint256 value) internal {
        bytes memory approvalCall = abi.encodeWithSelector(token.approve.selector, spender, value);

        if (!_callOptionalReturnBool(token, approvalCall)) {
            _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, 0));
            _callOptionalReturn(token, approvalCall);
        }
    }

    /**
     * @dev Use a ERC-2612 signature to set the `owner` approval toward `spender` on `token`.
     * Revert on invalid signature.
     */
    function safePermit(
        IERC20Permit token,
        address owner,
        address spender,
        uint256 value,
        uint256 deadline,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) internal {
        uint256 nonceBefore = token.nonces(owner);
        token.permit(owner, spender, value, deadline, v, r, s);
        uint256 nonceAfter = token.nonces(owner);
        require(nonceAfter == nonceBefore + 1, "SafeERC20: permit did not succeed");
    }

    /**
     * @dev Imitates a Solidity high-level call (i.e. a regular function call to a contract), relaxing the requirement
     * on the return value: the return value is optional (but if data is returned, it must not be false).
     * @param token The token targeted by the call.
     * @param data The call data (encoded using abi.encode or one of its variants).
     */
    function _callOptionalReturn(IERC20 token, bytes memory data) private {
        // We need to perform a low level call here, to bypass Solidity's return data size checking mechanism, since
        // we're implementing it ourselves. We use {Address-functionCall} to perform this call, which verifies that
        // the target address contains contract code and also asserts for success in the low-level call.

        bytes memory returndata = address(token).functionCall(data, "SafeERC20: low-level call failed");
        require(returndata.length == 0 || abi.decode(returndata, (bool)), "SafeERC20: ERC20 operation did not succeed");
    }

    /**
     * @dev Imitates a Solidity high-level call (i.e. a regular function call to a contract), relaxing the requirement
     * on the return value: the return value is optional (but if data is returned, it must not be false).
     * @param token The token targeted by the call.
     * @param data The call data (encoded using abi.encode or one of its variants).
     *
     * This is a variant of {_callOptionalReturn} that silents catches all reverts and returns a bool instead.
     */
    function _callOptionalReturnBool(IERC20 token, bytes memory data) private returns (bool) {
        // We need to perform a low level call here, to bypass Solidity's return data size checking mechanism, since
        // we're implementing it ourselves. We cannot use {Address-functionCall} here since this should return false
        // and not revert is the subcall reverts.

        (bool success, bytes memory returndata) = address(token).call(data);
        return
            success && (returndata.length == 0 || abi.decode(returndata, (bool))) && Address.isContract(address(token));
    }
}

// File: @openzeppelin/contracts/utils/Address.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (utils/Address.sol)

pragma solidity ^0.8.1;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     *
     * Furthermore, `isContract` will also return true if the target contract within
     * the same transaction is already scheduled for destruction by `SELFDESTRUCT`,
     * which only has an effect at the end of a transaction.
     * ====
     *
     * [IMPORTANT]
     * ====
     * You shouldn't rely on `isContract` to protect against flash loan attacks!
     *
     * Preventing calls from contracts is highly discouraged. It breaks composability, breaks support for smart wallets
     * like Gnosis Safe, and does not provide security since it can be circumvented by calling from a contract
     * constructor.
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize/address.code.length, which returns 0
        // for contracts in construction, since the code is only stored at the end
        // of the constructor execution.

        return account.code.length > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://consensys.net/diligence/blog/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.8.0/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(address target, bytes memory data, uint256 value) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verify that a low level call to smart-contract was successful, and revert (either by bubbling
     * the revert reason or using the provided one) in case of unsuccessful call or if target was not a contract.
     *
     * _Available since v4.8._
     */
    function verifyCallResultFromTarget(
        address target,
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        if (success) {
            if (returndata.length == 0) {
                // only check isContract if the call was successful and the return data is empty
                // otherwise we already know that it was a contract
                require(isContract(target), "Address: call to non-contract");
            }
            return returndata;
        } else {
            _revert(returndata, errorMessage);
        }
    }

    /**
     * @dev Tool to verify that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason or using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            _revert(returndata, errorMessage);
        }
    }

    function _revert(bytes memory returndata, string memory errorMessage) private pure {
        // Look for revert reason and bubble it up if present
        if (returndata.length > 0) {
            // The easiest way to bubble the revert reason is using memory via assembly
            /// @solidity memory-safe-assembly
            assembly {
                let returndata_size := mload(returndata)
                revert(add(32, returndata), returndata_size)
            }
        } else {
            revert(errorMessage);
        }
    }
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: contracts/BountyV1.sol



// SPDX-License-Identifier: UNLICENSED
pragma solidity 0.8.18;

import "@openzeppelin/contracts/access/Ownable.sol";
import "@openzeppelin/contracts/token/ERC20/utils/SafeERC20.sol";
import "@openzeppelin/contracts/token/ERC20/extensions/ERC20Burnable.sol";

/**
    @title Bounty Contract Version 1

    @notice This contract allows for the creation, funding, and completion of blockchain intelligence
    bounties in a decentralized manner.

    @dev This contract is intended to be used with the ARKM token.

    Users can submit bounties by putting up ARKM, and other users can complete the bounties by
    staking ARKM to submit IDs that correspond to solutions.

    Submission IDs are not submitted raw, but instead hashed with the submitter's address to
    prevent front-running. The IDs are effectively secrets that only the submitters know on
    submission.

    bountyIDs and submissionIDs correspond to off-chain records that detail the bounty and contain
    solutions. For v1, these records live on the Arkham platform. In the future there could be any
    number of secondary services that host these records.

    Once submissions are made, only a set of approver addresses can approve or reject them -- NOT
    the posters of the bounty. Approver addresses can be added or removed by the the contract
    owner.

    Once a bounty is funded, it cannot be closed until after its expiration. Bounties also must
    remain open until any submission to the bounty has been approved or rejected. Once a bounty is
    closed, it cannot be reopened. If a funder closes their bounty, they receive the amount of the
    bounty back. (Bounty funders can receive a maximum of 100% of their initial funding back. if
    there is excess funding from rejected submissions, it is accrued as fees.)

    The contract owner sets the initial submission stake, maker fee, and taker fee.

    A maker fee is charged when a bounty is funded. A taker fee is charged when a submission is
    paid out. These fees are in basis points, e.g. 100 basis points = 1%. Fees are disbursed at the
    discretion of the owner. A fraction of fees are burned, and the rest are withdrawn to an
    address set at contract creation and changeable by the owner.

    If a submission is approved, the submitter receives the bounty amount plus their stake back,
    less a taker fee. If a submission is rejected, the submitter does NOT receive their stake back,
    and the bounty amount increases by the amount staked. Only one submission may be active for a
    given bounty at a time.

    Delivery of the information corresponding to the accepted submission to the funder is handled
    off-chain.
 */
contract BountyV1 is Ownable {
    address private immutable _arkm;

    uint256 private _submissionStake;
    uint256 private _makerFee;
    uint256 private _takerFee;
    bool    private _acceptingBounties;
    uint256 private _accruedFees;
    address private _feeReceiverAddress;

    uint64 private immutable _bountyDuration;

    uint64  private constant _MAX_BPS = 10000;

    /// @dev Bounty ID --> Bounty
    mapping(uint256 => Bounty) private _bounties;

    /// @dev Bounty ID --> Submission. There can only be one submission per bounty at a time.
    mapping(uint256 => Submission) private _submissions;

    /// @dev Payload --> bool. Tracks rejected payloads.
    mapping(uint256 => bool) private _rejectedPayloads;

    /// @dev Approver address --> is approver
    mapping(address => bool) private _approvers;

    /// @notice Struct representing a bounty
    /// @dev No submissions may be posted to the bounty after the unlock time. If there is an
    /// active submission when the unlock time is reached, it may still be approved after the
    /// unlock time. The bounty is only considered closed when it is past the expiration AND there
    /// is no active submission. The ID corresponds to a record kept on the Arkham platform - or in
    /// the future, on any number of secondary serivices.
    struct Bounty {
        uint256 amount;
        uint256 initialAmount;
        uint64 expiration;
        address funder;
        bool closed;
    }

    /// @notice Struct representing a submission to a bounty
    /// @dev The payload is a hash of the submission ID and the submitter's address to prevent
    /// front-running. This ID corresponds to a record on the Arkham platform - or, in the future,
    /// any secondary service that bounty-approvers can use to verify the submission.
    struct Submission {
        bytes32 payload;
        address submitter;
        uint256 stake;
    }

    /// @notice Emitted when a bounty is funded
    /// @param bountyID The ID of the funded bounty
    /// @param funder The address of the funder
    /// @param initialValue The initial value of the bounty
    /// @param expiration The unlock time of the bounty
    event FundBounty (
        uint256 indexed bountyID,
        address indexed funder,
        uint256 initialValue,
        uint64  expiration
    );

    /// @notice Emitted when a submission is made for a bounty
    /// @param bountyID The ID of the bounty for which the submission is made
    /// @param submitter The address of the submitter
    /// @param stake The stake of the submission
    /// @param payload The payload of the submission
    event FundSubmission (
        uint256 indexed bountyID,
        address indexed submitter,
        uint256 stake,
        bytes32 payload
    );

    /// @notice Emitted when a submission for a bounty is rejected
    /// @param bountyID The ID of the bounty for which the submission is rejected
    /// @param submitter The address of the submitter
    /// @param stake The stake of the rejected submission
    /// @param payload The payload of the rejected submission
    event RejectSubmission (
        uint256 indexed bountyID,
        address indexed submitter,
        uint256 stake,
        bytes32 payload
    );

    /// @notice Emitted when a submission for a bounty is approved
    /// @param bountyID The ID of the bounty for which the submission is approved
    /// @param submitter The address of the submitter
    /// @param stake The stake of the approved submission
    /// @param payload The payload of the approved submission
    /// @param payoutToSubmitter The payout to the submitter
    event SubmissionApproved (
        uint256 indexed bountyID,
        address indexed submitter,
        uint256 stake,
        bytes32 payload,
        uint256 payoutToSubmitter
    );

    /// @notice Emitted when a bounty is closed
    /// @param bountyID The ID of the closed bounty
    /// @param funder The address of the funder
    /// @param payoutToFunder The payout to the funder
    /// @param excessFromStaking The excess fees
    event CloseBounty (
        uint256 indexed bountyID,
        address indexed funder,
        address indexed closedBy,
        uint256 payoutToFunder,
        uint256 excessFromStaking
    );

    /// @notice Emitted when an account is granted approver status
    /// @param account The account granted approver status
    event GrantApprover (
        address indexed account
    );

    /// @notice Emitted when an account has its approver status revoked
    /// @param account The account that had its approver status revoked
    event RevokeApprover (
        address indexed account
    );

    /// @notice Emitted when the maker fee is set
    /// @param newFee The new maker fee
    /// @param oldFee The old maker fee
    event SetMakerFee (
        uint256 newFee,
        uint256 oldFee
    );

    /// @notice Emitted when the taker fee is set
    /// @param newFee The new taker fee
    /// @param oldFee The old taker fee
    event SetTakerFee (
        uint256 newFee,
        uint256 oldFee
    );

    /// @notice Emitted when the submission stake is set
    /// @param newStake The new submission stake
    /// @param oldStake The old submission stake
    event SetSubmissionStake (
        uint256 newStake,
        uint256 oldStake
    );

    /// @notice Emitted when the contract stops accepting bounties
    event CloseBountySubmissions ();

    /// @notice Emitted when accrued fees are withdrawn
    /// @param amount The amount of accrued fees withdrawn
    event WithdrawFees (
        uint256 amount
    );

    /// @notice Contract constructor that sets initial values
    /// @param arkmAddress Address of the ERC20 token to be used (ARKM)
    /// @param initialSubmissionStake Initial stake required for a submission, e.g. 10 e18
    /// @param initialMakerFee Initial fee for creating a bounty, in basis points
    /// @param initialTakerFee Initial fee for completing a bounty, in basis points
    /// @param bountyDuration Duration of a bounty, in days, i.e. time until expiration
    /// @dev We check to make sure the arkmAddress implements ERC20Burnable so we can burn fees.
    /// This also makes sure the address is a valid ERC20 token.
    constructor(address arkmAddress, uint256 initialSubmissionStake, uint256 initialMakerFee, uint256 initialTakerFee, uint64 bountyDuration, address feeReceiverAddress) {
        require(initialMakerFee <= _MAX_BPS, "BountyV1: maker fee must be <= 10000");
        require(initialTakerFee <= _MAX_BPS, "BountyV1: taker fee must be <= 10000");
        require(feeReceiverAddress != address(0), "BountyV1: fee receiver address cannot be 0x0");
        require(bountyDuration <= 36500, "BountyV1: bounty duration must be <= 36500 days");

        try ERC20Burnable(arkmAddress).totalSupply() returns (uint256) {
            _arkm = arkmAddress;
        } catch {
            revert("BountyV1: provided token address does not implement ERC20Burnable");
        }

        _submissionStake = initialSubmissionStake;
        _makerFee = initialMakerFee;
        _takerFee = initialTakerFee;
        _acceptingBounties = true;
        _bountyDuration = bountyDuration;
        _feeReceiverAddress = feeReceiverAddress;
    }

    /// @return The amount of ARKM accrued from fees
    function accruedFees() public view virtual returns (uint256) {
        return _accruedFees;
    }

    /// @return The address of the ERC20 token to be used
    function arkm() public view virtual returns (address) {
        return _arkm;
    }

    /// @return The fee for creating a bounty, in basis points
    function makerFee() public view virtual returns (uint256) {
        return _makerFee;
    }

    /// @return The fee for completing a bounty, in basis points
    function takerFee() public view virtual returns (uint256) {
        return _takerFee;
    }

    /// @return The stake required for a submission, in value of the ERC20 token
    function submissionStake() public view virtual returns (uint256) {
        return _submissionStake;
    }

    /// @return The duration of a bounty, in days
    function bountyDurationDays() public view virtual returns (uint64) {
        return _bountyDuration;
    }

    /// @param bounty The int representation of the UUID of the bounty
    function funder(uint256 bounty) public view virtual returns (address) {
        return _bounties[bounty].funder;
    }

    /// @param bounty The ID of the bounty
    /// @return The amount of the bounty
    function amount(uint256 bounty) public view virtual returns (uint256) {
        return _bounties[bounty].amount;
    }

    /// @param bounty The ID of the bounty
    /// @return The initial amount of the bounty
    function initialAmount(uint256 bounty) public view virtual returns (uint256) {
        return _bounties[bounty].initialAmount;
    }

    /// @param bounty The ID of the bounty
    /// @return The expiration time of the bounty
    function expiration(uint256 bounty) public view virtual returns (uint64) {
        return _bounties[bounty].expiration;
    }

    /// @param bounty The ID of the bounty
    /// @return Whether the bounty is closed or not
    /// @dev 'closed' is only true once a submission has been approved OR the bounty has been
    /// closed by the funder, which can only happen after the unlock time when there is no active
    /// submission.
    function closed(uint256 bounty) public view virtual returns (bool) {
        return _bounties[bounty].closed;
    }

    /// @param payload The payload of a submission
    /// @return Whether the payload has been rejected or not
    /// @dev This is how we keep track of which submissions have been rejected.
    function rejectedPayload(bytes32 payload) public view virtual returns (bool) {
        return _rejectedPayloads[uint256(payload)] == true;
    }

    /// @param account The account to check
    /// @return Whether the account is an approver or not
    function approver(address account) public view virtual returns (bool) {
        return _approvers[account];
    }

    /// @param bounty The ID of the bounty the submission corresponds to
    /// @param submission The ID of the submission
    /// @return Whether the submission is approved or not
    function approvedSubmission(uint256 bounty, uint256 submission) public view virtual returns (bool) {
        return isValidCurrentSubmission(bounty, submission) && closed(bounty);
    }

    /// @param bounty The ID of the bounty
    /// @return The address of the submitter of the bounty
    function submitter(uint256 bounty) public view virtual returns (address) {
        return _submissions[bounty].submitter;
    }

    /// @param bounty The ID of the bounty
    /// @return The amount of the ERC20 staked in the submission of the bounty
    /// @dev If there is no submission, this will return 0.
    function submissionStaked(uint256 bounty) public view virtual returns (uint256) {
        return _submissions[bounty].stake;
    }

    /// @return Whether the contract is accepting bounties or not
    function acceptingBounties() public view virtual returns (bool) {
        return _acceptingBounties;
    }

    /// @param value The value to calculate the fee from
    /// @param maker Whether the fee is for creating a bounty or not
    /// @return The calculated fee
    /// @dev The 10000 accounts for denomination in basis points.
    function fee(uint256 value, bool maker) public view virtual returns (uint256) {
        return value * (maker ? _makerFee : _takerFee) / 10000;
    }

    /// @notice Validates a submission for a bounty
    /// @param bounty The ID of the bounty the submission corresponds to
    /// @param submission The ID of the submission
    /// @return Whether the submission is valid or not
    /// @dev The submissionID x submitterAddress hash must match the payload of the submission.
    function isValidCurrentSubmission(uint256 bounty, uint256 submission) public view virtual returns (bool) {
        return keccak256(abi.encodePacked(submission, submitter(bounty))) == _submissions[bounty].payload;
    }

    /// @notice Modifier to require that the caller is an approver
    modifier onlyApprover() {
        require(approver(_msgSender()), "BountyV1: caller is not approver");
        _;
    }

    /// @notice Funds a bounty
    /// @param bounty The ID of the bounty to fund
    /// @param _amount The amount of the ERC20 to fund the bounty with
    /// @dev Additional bounty information like name and description is stored on an off-chain
    /// platform, e.g. Arkham. In the current implementation, Bounty records are generated
    /// on the Arkham platform and the bountyID is the int representation of the UUID of those
    /// records.
    function fundBounty(uint256 bounty, uint256 _amount) external {
        require(_acceptingBounties, "BountyV1: contract no longer accepting bounties");
        require(_amount > 0, "BountyV1: amount must be > 0");
        require(amount(bounty) == 0, "BountyV1: bounty already funded");
        require(_amount - fee(_amount, true) > _submissionStake, "BountyV1: amount after fee must be at least the submission stake");

        // Should not allow funding a bounty that is closed. However that is covered by the
        // fact that you can not fund an already funded bounty and you can not close an
        // unfunded bounty.

        _accruedFees += fee(_amount, true);

        uint64 _expiration = uint64(block.timestamp + _bountyDuration * 1 days);

        _bounties[bounty] = Bounty({
            amount: _amount - fee(_amount, true),
            initialAmount: _amount - fee(_amount, true),
            expiration: _expiration,
            funder: _msgSender(),
            closed: false
        });
        SafeERC20.safeTransferFrom(IERC20(_arkm), _msgSender(), address(this), _amount);

        emit FundBounty(
            bounty,
            _msgSender(),
            _amount,
            _expiration
        );
    }

    /// @notice Closes a bounty and returns the funds to the funder
    /// @param bounty The ID of the bounty to close
    /// @dev The bounty may be closed by anyone after the unlock time when there is no active
    /// submission. Funds are returned to the funder. Approvers can close bounties before expiration.
    function closeBounty(uint256 bounty) external {
        require(amount(bounty) > 0, "BountyV1: bounty not funded");
        require(expiration(bounty) <= block.timestamp || approver(_msgSender()), "BountyV1: only approvers can close before expiration");
        require(_submissions[bounty].submitter == address(0), "BountyV1: has active submission");
        require(!closed(bounty), "BountyV1: bounty already closed");

        /// @dev If there have been rejected submissions, the funder will receive up to their full
        /// initial funding amount. Staked funds from rejected submissions are accrued as fees.
        uint256 excessFromStaking = _bounties[bounty].amount - _bounties[bounty].initialAmount;

        _bounties[bounty].closed = true;
        SafeERC20.safeTransfer(IERC20(_arkm), _bounties[bounty].funder, _bounties[bounty].initialAmount);

        _accruedFees += excessFromStaking;

        emit CloseBounty(
            bounty,
            _bounties[bounty].funder,
            _msgSender(),
            _bounties[bounty].initialAmount,
            excessFromStaking
        );
    }

    /// @notice Makes a submission for a bounty by staking the ERC20
    /// @param bounty The ID of the bounty to make a submission to
    /// @param payload The payload of the submission, used to validate the sender's address
    /// @dev The payload is a hash of the submission ID and the submitter's address to prevent
    /// front-running. All submissions must provide a stake, which is returned if the submission
    /// is approved (less fees) or forfeit (and added to the bounty) if the submission is rejected.
    /// This is to prevent spamming of submissions.
    function makeSubmission(uint256 bounty, bytes32 payload) external {
        require(amount(bounty) > 0, "BountyV1: bounty not funded");
        // Not allowed to make submission on closed bounty but it is covered by the fact that
        // you can not close a bounty that is not expired.
        require(expiration(bounty) > block.timestamp, "BountyV1: bounty expired");
        require(submitter(bounty) == address(0), "BountyV1: submission already made");
        require(!rejectedPayload(payload), "BountyV1: payload rejected");
        require(!approver(_msgSender()), "BountyV1: approvers cannot submit");

        _submissions[bounty] = Submission({
            payload: payload,
            submitter: _msgSender(),
            stake: _submissionStake
        });
        SafeERC20.safeTransferFrom(IERC20(_arkm), _msgSender(), address(this), _submissionStake);

        emit FundSubmission(
            bounty,
            _msgSender(),
            _submissionStake,
            payload
        );
    }

    /// @notice Approves and pays out for a submission to a bounty
    /// @param bounty The ID of the bounty whose submission to approve
    /// @param submission The ID of the submission to approve
    /// @dev The bounty is paid out to the address that submitted the approved submission, plus
    /// their initial stake, less fees. The bounty is closed and will no longer receive
    /// submissions. Off-chain, the information corresponding to the submission ID is delivered to
    /// the funder.
    function approveSubmission(uint256 bounty, uint256 submission) external onlyApprover {
        require(submitter(bounty) != address(0), "BountyV1: no submission not made");
        // Approving a submission on a closed bounty is not allowed but handled by check that makes
        // it impossible to submit to a closed bounty and the check that makes it impossible to close
        // a bounty that has an active submission.
        require(isValidCurrentSubmission(bounty, submission), "BountyV1: invalid submission");
        require(submitter(bounty) != _msgSender(), "BountyV1: cannot approve own submission");

        /// @dev If there have been other rejected submissions, their stake will be reflected in
        /// the bounty amount.
        uint256 _amount = amount(bounty);
        uint256 _stake = submissionStaked(bounty);
        _accruedFees += fee(_amount, false);

        _bounties[bounty].closed = true;

        uint256 payout = _amount + _stake - fee(_amount, false);
        SafeERC20.safeTransfer(IERC20(_arkm), submitter(bounty), payout);

        emit SubmissionApproved(
            bounty,
            submitter(bounty),
            _stake,
            _submissions[bounty].payload,
            payout
        );
    }

    /// @notice Rejects a submission for a bounty
    /// @param bounty The ID of the bounty whose submission to reject
    /// @dev Only one submission may exist for a bounty at a time. Once the submission is rejected,
    /// another submission may be made, as long as it is not past the bounty's unlock time. The
    /// stake of the rejected submission is forfeit and added to the bounty amount (less fees).
    function rejectSubmission(uint256 bounty) external onlyApprover {
        require(submitter(bounty) != address(0), "BountyV1: submission not made");
        // Should not be able to reject submissions on closed bounties but you are not allowed
        // to close a bounty that has an active submission so it is covered.
        // Neither should you be able to reject a payload twice, but after it is rejected it gets
        // removed from the _subbmission mapping and thus fall under the "submission not made".

        _accruedFees += fee(_submissions[bounty].stake, true);
        _bounties[bounty].amount += _submissions[bounty].stake - fee(_submissions[bounty].stake, true);
        _rejectedPayloads[uint256(_submissions[bounty].payload)] = true;

        emit RejectSubmission(
            bounty,
            submitter(bounty),
            _submissions[bounty].stake,
            _submissions[bounty].payload
        );
        // Remove the submission that has been rejected.
        delete _submissions[bounty];
    }

    /// @notice Grants approver status to an account
    /// @param account The account to grant approver status to
    /// @dev There can be any number of approvers. Only one approver is required to approve a
    /// submission.
    function grantApprover(address account) external onlyOwner {
        require(!approver(account), "BountyV1: already approver");
        _approvers[account] = true;

        emit GrantApprover(
            account
        );
    }

    /// @notice Revokes approver status from an account
    /// @param account The account to revoke approver status from
    function revokeApprover(address account) external onlyOwner {
        require(approver(account), "BountyV1: not approver");
        _approvers[account] = false;

        emit RevokeApprover(
            account
        );
    }


    /// @notice Sets a new maker fee
    /// @param newFee The new maker fee, in basis points
    function setMakerFee(uint256 newFee) external onlyOwner {
        require(newFee <= _MAX_BPS, "BountyV1: maker fee must be <= 100%");
        uint256 _oldFee = _makerFee;
        _makerFee = newFee;

        emit SetMakerFee(
            newFee,
            _oldFee
        );
    }

    /// @notice Sets a new taker fee
    /// @param newFee The new taker fee, in basis points
    function setTakerFee(uint256 newFee) external onlyOwner {
        require(newFee <= _MAX_BPS, "BountyV1: taker fee must be <= 100%");
        uint256 _oldFee = _takerFee;
        _takerFee = newFee;

        emit SetTakerFee(
            newFee,
            _oldFee
        );
    }

    /// @notice Sets a new submission stake
    /// @param newStake The new submission stake, in value of the ERC20 token e.g. 10 e18
    function setSubmissionStake(uint256 newStake) external onlyOwner {
        uint256 _oldStake = _submissionStake;
        _submissionStake = newStake;

        emit SetSubmissionStake(
            newStake,
            _oldStake
        );
    }

    /// @notice Sets a new fee receiver address
    /// @param feeReceiverAddress The new fee receiver address
    /// @dev This is the address that will receive fees when there is a burn.
    function setFeeReceiverAddress(address feeReceiverAddress) external onlyOwner {
        require(feeReceiverAddress != address(0), "BountyV1: fee receiver address cannot be 0x0");
        _feeReceiverAddress = feeReceiverAddress;
    }

    /// @notice Prevent any further bounties from being created
    /// @dev This is how we will sunset the contract when we want to move to a new version. It's
    /// important that we continue to allow submissions to existing bounties until they all expire.
    /// Once all bounties have expired and we have approved or rejected all submissions, we can
    /// disburse any remaining fees.
    function stopAcceptingBounties() external onlyOwner {
        _acceptingBounties = false;

        emit CloseBountySubmissions();
    }

    /// @notice Burn fees and widthraw the rest
    /// @dev This can be called periodically by the owner. The burn rate is fixed. The burn
    /// fraction is in basis points.
    function withdrawFees() external onlyOwner {
        uint256 fees = _accruedFees;
        _accruedFees = 0;
        SafeERC20.safeTransfer(IERC20(_arkm), _feeReceiverAddress, fees);

        emit WithdrawFees(
            fees
        );
    }
}